



# An alternative numerical solution for the Orr–Sommerfeld problem

Antonio Barletta<sup>a</sup> , Pedro Vayssière Brandão<sup>b</sup> , Michele Celli<sup>c</sup> 

Department of Industrial Engineering, Alma Mater Studiorum Università di Bologna, Viale Risorgimento 2, 40136 Bologna, Italy

Received: 15 December 2023 / Accepted: 7 January 2024  
© The Author(s) 2024

**Abstract** A simple tutorial for the numerical solution of the Orr–Sommerfeld problem is presented. The aim is the determination of the neutral stability threshold to the linear instability for the Poiseuille flow within a plane parallel channel. Such an instability is hydrodynamic in nature as it is driven by the advection term in the local momentum balance equation. The numerical solution is achieved by a spectral method. The basis functions are chosen as Chebyshev polynomials multiplied by a suitable overall factor meant to allow the fulfilment of the boundary conditions. The details of the numerical code implemented via the *Mathematica* software environment are also provided. A comparison with other numerical solutions of the same problem implemented via the tau-method is presented.

## 1 Introduction

The linear stability analysis of plane Poiseuille flow is a classical problem in fluid dynamics. Plane Poiseuille flow refers to the laminar fully-developed flow of an incompressible Newtonian fluid between two parallel plates driven by a constant pressure gradient. The governing equations for the problem are the Navier-Stokes equation, which describes the local momentum balance, and the continuity equation, which describes the local mass balance and constrains the velocity field to be solenoidal. For the linear stability analysis, small perturbations around the base flow are considered. A detailed account of this analysis and its physical background can be found in several books such as [1] or [2].

The core of the linear stability analysis is the fourth-order differential eigenvalue problem well-known as the Orr–Sommerfeld problem. The critical Reynolds number for the linear instability of plane Poiseuille flow is approximately  $Re_c \approx 5772$  [2]. Below this critical Reynolds number, the flow is linearly stable and, above it, the flow is unstable. It must be stressed that linear stability analysis provides insights into the onset of instability, but it may not fully capture the complete mechanisms of transition, which often involve nonlinear effects and the development of finite-amplitude disturbances. On this topic, there are recent interesting investigations carried out in [3, 4] and [5].

The aim of this paper is to present a tutorial that provides all the details to achieve an accurate numerical solution for the Orr–Sommerfeld problem relative to the linear stability of the plane Poiseuille flow in a channel. The pedagogical approach adopted in this paper is meant to provide the reader with all the coding details needed for reproducing the numerical data presented. For the sake of brevity, we decided to refer the reader to textbooks for the underlying physical problem, as it is a cornerstone topic of fluid mechanics widely and effectively presented by many authors [1, 2]. Likewise, the basic results, such as the focus on the two-dimensional modes established by Squire's theorem, leading to the formulation of the Orr–Sommerfeld eigenvalue problem can be found, for instance, in [1] or [2]. Our scope is limited to the numerical solution of the linear stability eigenvalue problem which is, in most cases, not described in all its details. Furthermore, we have chosen to adopt a numerical method which differs from that universally employed for this type of differential eigenvalue problems, namely the tau-method [6–9]. In the formulation of the method, we got inspired by the excellent paper from Fujimura and Kelly [10] where, however, not much details were provided regarding the solution technique. Step-by-step, the whole numerical procedure is described and implemented by providing its coding in the *Mathematica* language [11].

## 2 Definition of the problem

We consider the Orr–Sommerfeld eigenvalue problem defining the linear instability of the Poiseuille flow in a plane channel. Further details can be found in such papers as [6, 10, 12] as well as in textbooks such as [1] or [2].

---

Antonio Barletta, Pedro Vayssière Brandão and Michele Celli have contributed equally to this work.

<sup>a</sup> e-mail: [antonio.barletta@unibo.it](mailto:antonio.barletta@unibo.it) (corresponding author)

<sup>b</sup> e-mail: [pedro.vayssiere2@unibo.it](mailto:pedro.vayssiere2@unibo.it)

<sup>c</sup> e-mail: [michele.celli3@unibo.it](mailto:michele.celli3@unibo.it)

The base dimensionless velocity profile for the flow along the  $x$  direction is given by

$$U(z) = 1 - z^2 \quad -1 \leq z \leq 1, \quad (1)$$

with  $z$  the transverse dimensionless coordinate.

The Orr–Sommerfeld problem associated with the base velocity profile (1) is formulated as

$$\begin{aligned} \frac{1}{Re} [\phi''''(z) - 2\alpha^2 \phi''(z) + \alpha^4 \phi(z)] - i\alpha U(z) [\phi''(z) - \alpha^2 \phi(z)] + i\alpha U''(z) \phi(z) \\ = c \{-i\alpha [\phi''(z) - \alpha^2 \phi(z)]\}, \end{aligned} \quad (2)$$

$$\phi(\pm 1) = 0 \quad \phi'(\pm 1) = 0, \quad (3)$$

where  $Re$  is the Reynolds number for the Poiseuille flow (1),  $\alpha$  and  $c$  are the real wavenumber and the complex phase speed of the streamfunction disturbances, while the primes denote derivatives with respect to  $z$ . We mention that the dimensionless velocity  $U(z)$  has been scaled with the maximum base flow value, occurring at  $z = 0$ , while the dimensionless coordinate  $z$  has been scaled by the channel half-width. Such reference scales for the velocity and length are those adopted for the definition of the Reynolds number. The eigenfunction  $\phi(z)$  defines the streamfunction perturbation relative to the base flow given by (1),

$$\Phi(x, z, t) = \phi(z) e^{i\alpha(x - ct)} \quad \alpha \in \mathbb{R} \quad c \in \mathbb{C}, \quad (4)$$

where  $t$  is time. With the eigenvalue given by  $c = c_r + i c_i$ , the transition to linear instability happens when  $c_i$  changes from negative to positive.

### 3 Numerical solution

The solution of (2, 3) is obtained numerically as follows. Let us express the eigenfunction  $\phi(z)$  as

$$\phi(z) = \sum_{n=0}^N a_n \psi_n(z). \quad (5)$$

Here,  $\psi_n(z)$  are basis functions satisfying the boundary conditions (3). A convenient choice is suggested in [10],

$$\psi_n(z) = [U(z)]^2 T_{2n}(z), \quad (6)$$

where  $T_{2n}(z)$ , with  $-1 \leq z \leq 1$ , are the Chebyshev polynomials defined as

$$T_{2n}(\cos \theta) = \cos(2n\theta) \quad \theta \in \mathbb{R}. \quad (7)$$

A sketch of the first five basis functions  $\psi_n(z)$  is displayed in Fig. 1.

We introduce the shorthand notation,

$$\langle f, g \rangle = \int_{-1}^1 f(z) g(z) dz. \quad (8)$$

Then, (2) can be rewritten as

$$\sum_{n=0}^N A_{mn} a_n = c \sum_{n=0}^N B_{mn} a_n, \quad (9)$$


where

$$\begin{aligned} A_{mn} &= \frac{1}{Re} [\langle \psi_m, \psi_n'''' \rangle - 2\alpha^2 \langle \psi_m, \psi_n'' \rangle + \alpha^4 \langle \psi_m, \psi_n \rangle] - i\alpha \langle \psi_m, U \psi_n'' \rangle \\ &\quad + i\alpha^3 \langle \psi_m, U \psi_n \rangle - 2i\alpha \langle \psi_m, \psi_n \rangle, \\ B_{mn} &= -i\alpha [\langle \psi_m, \psi_n'' \rangle - \alpha^2 \langle \psi_m, \psi_n \rangle]. \end{aligned} \quad (10)$$

In matrix notation, by using  $A$  for the  $(N+1) \times (N+1)$  matrix with elements  $A_{mn}$ ,  $B$  for the  $(N+1) \times (N+1)$  matrix with elements  $B_{mn}$  and  $a$  for the eigenvector  $(a_0, a_1, a_2, \dots, a_N)$ , we can rewrite (9) as the generalised eigenvalue problem

$$A \cdot a = c B \cdot a. \quad (11)$$

The matrix  $A$  depends on the parameters  $(\alpha, Re)$ , while  $B$  depends only on  $\alpha$ .

**Fig. 1** Plots of the first five basis functions  $\psi_n(z)$ 

#### 4 Numerical code

The *Mathematica* coding of the generalised eigenvalue problem (11), is obtained by defining the matrices  $A$  and  $B$  as follows:

```

NN = 35;

U[z_] := 1 - z^2;

p[n_, z_] := (U[z])^2 ChebyshevT[2 n, z];

Do[A0[n, m] = Integrate[psi[n, z] psi[m, z], {z, -1, 1}], {n, 0, NN}, {m, 0, NN}];

Do[A2[n, m] =
  Integrate[psi[n, z] D[psi[m, z], {z, 2}], {z, -1, 1}], {n, 0, NN}, {m, 0, NN}];

Do[A4[n, m] =
  Integrate[psi[n, z] D[psi[m, z], {z, 4}], {z, -1, 1}], {n, 0, NN}, {m, 0, NN}];

Do[AU0[n, m] =
  Integrate[psi[n, z] U[z] psi[m, z], {z, -1, 1}], {n, 0, NN}, {m, 0, NN}];

Do[AU2[n, m] =
  Integrate[psi[n, z] U[z] D[psi[m, z], {z, 2}], {z, -1, 1}],
  {n, 0, NN}, {m, 0, NN}];

AA[alpha_, RE_, M_] :=
  Table[1/RE (A4[n, m] - 2 alpha^2 A2[n, m] + alpha^4 A0[n, m]) -
    I alpha AU2[n, m] + I alpha^3 AU0[n, m] -
    2 I alpha A0[n, m], {n, 0, M}, {m, 0, M}];

BB[alpha_, M_] :=
  Table[- I alpha A2[n, m] + I alpha^3 A0[n, m], {n, 0, M}, {m, 0, M}];

```

provided that the order  $N$  of the approximation must be less than or equal to the prefixed value  $NN = 35$  and prescribed by the variable  $M$  in the definition of the matrices  $A$  and  $B$ . Then, the eigenvalues of the generalised problem (11) are determined, for the sample case where  $\alpha = 1.02055$ ,  $Re = 5772.222$ , with

```
SortBy[With[{alpha=1.02055,RE=5772.222,M=30},
Eigenvalues[{AA[alpha,RE,M],BB[alpha,M]}],N[Im[#]]&]//MatrixForm
```

Function `SortBy` serves to order the eigenvalues with an increasing imaginary part. Thus, the last element of the list is the eigenvalue  $c$  having the largest growth rate  $c_i$ . In this example, with  $M = 30$ , the output is

```
{0.0353791 - 74.4742 I},
{0.0488949 - 18.7072 I},
{0.0712486 - 8.38027 I},
{0.102163 - 4.7658 I},
{0.141183 - 3.09193 I},
{0.187533 - 2.18023 I},
{0.239897 - 1.62587 I},
{0.296326 - 1.25905 I},
{0.354548 - 0.998585 I},
{0.412618 - 0.802175 I},
{0.469341 - 0.646295 I},
{0.871279 - 0.562672 I},
{0.524189 - 0.517082 I},
{0.793675 - 0.445723 I},
{0.576476 - 0.406115 I},
{0.739504 - 0.347956 I},
{0.604151 - 0.320122 I},
{0.652539 - 0.303308 I},
{0.694802 - 0.286029 I},
{0.442918 - 0.270595 I},
{0.733216 - 0.264204 I},
{0.552351 - 0.24417 I},
{0.769484 - 0.228049 I},
{0.227726 - 0.208297 I},
{0.806383 - 0.191563 I},
{0.84329 - 0.155121 I},
{0.403557 - 0.152351 I},
{0.880183 - 0.118677 I},
{0.917063 - 0.0822251 I},
{0.95393 - 0.0457648 I},
{0.264001 + 3.01804*10^-10 I}
```

It is the last eigenvalue  $c$  in the list that displays  $c_i > 0$  and, hence, instability. In fact, the critical condition for instability is, according to [10],

$$(Re_c, \alpha_c, c_{rc}) = (5772.221816, 1.02054744, 0.2640002605). \quad (12)$$

As a cross-check, with  $\alpha = 1.02055$ ,  $Re = 5772.221$ , the eigenvalues are evaluated with

```
SortBy[With[{alpha=1.02055,RE=5772.221,M=30},
Eigenvalues[{AA[alpha,RE,M],BB[alpha,M]}],N[Im[#]]&]//MatrixForm
```

again with  $M = 30$ . This yields the output

```

{0.0353791 - 74.4742 I},
{0.0488949 - 18.7072 I},
{0.0712486 - 8.38027 I},
{0.102163 - 4.7658 I},
{0.141183 - 3.09193 I},
{0.187533 - 2.18023 I},
{0.239897 - 1.62587 I},
{0.296326 - 1.25905 I},
{0.354548 - 0.998585 I},
{0.412618 - 0.802175 I},
{0.469341 - 0.646295 I},
{0.871279 - 0.562672 I},
{0.524189 - 0.517082 I},
{0.793675 - 0.445723 I},
{0.576476 - 0.406115 I},
{0.739504 - 0.347956 I},
{0.604151 - 0.320123 I},
{0.652539 - 0.303308 I},
{0.694802 - 0.286029 I},
{0.442918 - 0.270595 I},
{0.733216 - 0.264204 I},
{0.552351 - 0.24417 I},
{0.769484 - 0.228049 I},
{0.227726 - 0.208297 I},
{0.806383 - 0.191563 I},
{0.84329 - 0.155122 I},
{0.403557 - 0.152351 I},
{0.880183 - 0.118677 I},
{0.917063 - 0.0822251 I},
{0.95393 - 0.0457648 I},
{0.264001 - 1.34681*10^-9 I}

```

As expected, all eigenvalues have  $c_i < 0$  in this case.

In the lists displayed above, only the last eigenvalue (that with a maximum  $c_i$ ) is relevant for determining the transition to linear instability. In order to get just that eigenvalue, one can employ function `Last`,

```

SortBy[With[{alpha = 1.02055, RE = 5772.222, M = 30},
  Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}]], N[Im[#]] &] // Last

```

with the output

```
0.264001 + 3.01804*10^-10 I
```

A fine tuning in the determination of the neutral stability threshold for a given  $\alpha$  can be obtained by gradually increasing  $Re$  with a `Do` loop,

```

Do[Print[NumberForm[RE, {8, 4}], ", ",
  SortBy[With[{alpha = 1.02055, M = 30},
    Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}]], N[Im[#]] &] //
  Last], {RE, 5772.2211, 5772.2225, 1/10000}]

```

The output is

```

5772.2211, 0.264001 -1.18194*10^-9 I
5772.2212, 0.264001 -1.01709*10^-9 I
5772.2213, 0.264001 -8.52245*10^-10 I
5772.2214, 0.264001 -6.87347*10^-10 I
5772.2215, 0.264001 -5.22487*10^-10 I
5772.2216, 0.264001 -3.5767*10^-10 I
5772.2217, 0.264001 -1.92765*10^-10 I
5772.2218, 0.264001 -2.794*10^-11 I
5772.2219, 0.264001 +1.36924*10^-10 I
5772.2220, 0.264001 +3.01804*10^-10 I
5772.2221, 0.264001 +4.66645*10^-10 I
5772.2222, 0.264001 +6.31505*10^-10 I
5772.2223, 0.264001 +7.96357*10^-10 I
5772.2224, 0.264001 +9.6124*10^-10 I
5772.2225, 0.264001 +1.12608*10^-9 I

```

with the transition to instability within  $Re = 5772.2218$  and  $Re = 5772.2219$ .

A direct evaluation of the neutral stability value of  $Re$  can be obtained by stopping the `Do` loop at the first iteration where  $c_i > 0$ . This is obtained with `Break[]`. Then, the code

```

With[{alpha = 1.02055, M = 30},
  Do[aa = (SortBy[Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}],
    N[Im[#]] &] // Last); R = RE; alphaalpha = alpha;
    If[Im[aa] > 0, Break[]], {RE, 5772.0, 10000.0, 0.001}];
  Print[{alphaalpha, R, Re[aa]}]]

```

yields the output

```
{1.02055, 5772.22, 0.264001}
```

where the first entry is  $\alpha$  and the third entry is  $c_r$ .

By varying the value of  $\alpha$ , one can now get the data needed for drawing the neutral stability curve in the  $(\alpha, Re)$  plane. This is achieved by running a `Do` loop over a varying  $\alpha$ ,

```

Do[With[{alpha = kk, M = 30},
  Do[aa = (SortBy[Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}],
    N[Im[#]] &] // Last); R = RE; alphaalpha = alpha;
    If[Im[aa] > 0, Break[]], {RE, 5772.0, 10000.0, 0.01}];
  Print[{alphaalpha, R, Re[aa]}, ", "], {kk, 1.015, 1.025, 0.001}]

```

so that we obtain the output


```

{1.015, 5775.55, 0.263324},
{1.016, 5774.47, 0.26345},
{1.017, 5773.60, 0.263574},
{1.018, 5772.94, 0.263697},
{1.019, 5772.49, 0.263817},
{1.020, 5772.26, 0.263936},
{1.021, 5772.25, 0.264053},
{1.022, 5772.46, 0.264168},
{1.023, 5772.90, 0.264281},
{1.024, 5773.57, 0.264391},
{1.025, 5774.48, 0.2645},

```

By collecting the neutral stability data in this way, one can draw the neutral stability curve (see Fig. 2).

**Fig. 2** Neutral stability curve



**5 Convergence test**

A convergence test of the method can be done by changing the order  $N$  of the approximation, namely the variable  $M$ , with the code

```
Do[Print["M = " NumberForm[M, {8, 4}], ", ", " ",
SortBy[With[{alpha = 1.02055, RE = 5773},
Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}]], N[Im[#]] &] //
Last], {M, 15, 35, 1}]
```

The output is

```
M = 15, 0.263995 -7.40786*10^-7 I
M = 16, 0.263994 +1.88333*10^-6 I
M = 17, 0.263994 +1.10721*10^-6 I
M = 18, 0.263994 +1.32944*10^-6 I
M = 19, 0.263994 +1.27151*10^-6 I
M = 20, 0.263994 +1.28515*10^-6 I
M = 21, 0.263994 +1.28225*10^-6 I
M = 22, 0.263994 +1.28277*10^-6 I
M = 23, 0.263994 +1.2827*10^-6 I
M = 24, 0.263994 +1.2827*10^-6 I
M = 25, 0.263994 +1.2827*10^-6 I
M = 26, 0.263994 +1.2827*10^-6 I
M = 27, 0.263994 +1.2827*10^-6 I
M = 28, 0.263994 +1.2827*10^-6 I
M = 29, 0.263994 +1.2827*10^-6 I
M = 30, 0.263994 +1.2827*10^-6 I
M = 31, 0.263994 +1.2827*10^-6 I
M = 32, 0.263994 +1.2827*10^-6 I
M = 33, 0.263994 +1.2827*10^-6 I
M = 34, 0.263994 +1.2827*10^-6 I
M = 35, 0.263994 +1.2827*10^-6 I
```

which shows that the output remains unchanged for every value of  $M$  larger than 22. It is only slightly less accurate when considering larger values of  $Re$  as, say,  $Re = 2 \cdot 10^4$ ,

**Table 1** The most unstable eigenvalue  $c$  for  $\alpha = 1$  and  $Re = 10000$

$N$	$c$ (Ref. [6])	$c$ (Ref. [12])	$c$ (present report)
13	0.23713751 + 0.00563644 <i>i</i>	0.23757046 + 0.00374610 <i>i</i>	0.23750572 + 0.00381438 <i>i</i>
14	0.23690887 + 0.00365516 <i>i</i>	0.23743315 + 0.00372248 <i>i</i>	0.23752121 + 0.00371453 <i>i</i>
16	0.23743315 + 0.00372248 <i>i</i>	0.23752392 + 0.00375328 <i>i</i>	0.23752262 + 0.00373679 <i>i</i>
19	0.23752676 + 0.00373427 <i>i</i>	0.23752595 + 0.00373908 <i>i</i>	0.23752669 + 0.00373963 <i>i</i>
22	0.23752670 + 0.00373982 <i>i</i>	0.23752651 + 0.00373968 <i>i</i>	0.23752648 + 0.00373967 <i>i</i>
23	–	0.23752648 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>
24	–	0.23752649 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>
25	0.23752648 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>
28	0.23752649 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>	0.23752649 + 0.00373967 <i>i</i>
31	0.23752649 + 0.00373967 <i>i</i>	–	0.23752649 + 0.00373967 <i>i</i>

```
Do[Print["M = " NumberForm[M, {8, 4}], ", ",
SortBy[With[{alpha = 1.095, RE = 20000},
Eigenvalues[{AA[alpha, RE, M], BB[alpha, M]}]], N[Im[#]] &] //
Last], {M, 15, 35, 1}]
```

The output is

```
M = 15, 0.215860 -0.00385505 I
M = 16, 0.215939 -0.00423504 I
M = 17, 0.216028 -0.00408384 I
M = 18, 0.215947 -0.00412313 I
M = 19, 0.215993 -0.00413506 I
M = 20, 0.215980 -0.00411736 I
M = 21, 0.215980 -0.00412696 I
M = 22, 0.215983 -0.00412320 I
M = 23, 0.215981 -0.00412413 I
M = 24, 0.215982 -0.00412410 I
M = 25, 0.215981 -0.00412398 I
M = 26, 0.215981 -0.00412407 I
M = 27, 0.215981 -0.00412403 I
M = 28, 0.215981 -0.00412404 I
M = 29, 0.215981 -0.00412404 I
M = 30, 0.215981 -0.00412404 I
M = 31, 0.215981 -0.00412404 I
M = 32, 0.215981 -0.00412404 I
M = 33, 0.215981 -0.00412404 I
M = 34, 0.215981 -0.00412404 I
M = 35, 0.215981 -0.00412404 I
```

An interesting convergence test is carried out for  $\alpha = 1$  and  $Re = 10000$  by a comparison with the table reported in [12], where the author compares his results with those reported in [6]. This comparison is displayed in Table 1. It reveals a quite satisfactory agreement and a convergence comparable or slightly better than that displayed by the solution reported in [12]. In fact, [12] declares  $N + 1 = 25$  terms to be solved for in order to attain 8 significant figures accuracy, while the present method yields the same result with  $N + 1 = 24$  terms. For the sake of completeness, the numerical solution in [6] requires  $N + 1 = 29$  terms to achieve the same accuracy.

### 6 Conclusions

A numerical solution of the Orr–Sommerfeld eigenvalue problem for the linear stability of the Poiseuille flow in a plane channel has been presented in detail. The solution is obtained by an expansion of the eigenfunctions in series of basis functions expressed in terms of Chebyshev polynomials. All the details of the numerical method, inspired by the study reported in [10], have been presented.



Furthermore, the coding of the numerical method in *Mathematica* has been described step-by-step. A pedagogical approach has been chosen in order to fill a gap in the literature where the details of the numerical method and its coding are in many cases omitted. The convergence in the numerical evaluation of the eigenvalues has been tested and compared with sample data reported by other authors. The comparison revealed that our numerical solution yields an accurate determination of the most unstable eigenvalue with a number of terms in the polynomial expansion less or equal to that reported in [6] and in [12]. Another advantage of the numerical solution presented here, with respect to other numerical solutions available in the literature mostly based on the tau-method, is its evidently simpler formulation. The reason is that the basis functions are chosen so that they satisfy the boundary conditions. The same method can be extended to other situations where there is a thermal coupling in the momentum balance such as in convection heat transfer problems [10], or in other hydrodynamic instability problems where velocity boundary conditions departing from the no-slip conditions are considered.

**Acknowledgements** The authors were supported by Alma Mater Studiorum Università di Bologna with grant RFO 2022.

**Funding** Open access funding provided by Alma Mater Studiorum - Università di Bologna within the CRUI-CARE Agreement.

**Open Access** This article is licensed under a Creative Commons Attribution 4.0 International License, which permits use, sharing, adaptation, distribution and reproduction in any medium or format, as long as you give appropriate credit to the original author(s) and the source, provide a link to the Creative Commons licence, and indicate if changes were made. The images or other third party material in this article are included in the article's Creative Commons licence, unless indicated otherwise in a credit line to the material. If material is not included in the article's Creative Commons licence and your intended use is not permitted by statutory regulation or exceeds the permitted use, you will need to obtain permission directly from the copyright holder. To view a copy of this licence, visit <http://creativecommons.org/licenses/by/4.0/>.

## References

1. P.J. Schmid, D.S. Henningson, *Stability and Transition in Shear Flows* (Springer, New York, 2001)
2. P.G. Drazin, W.H. Reid, *Hydrodynamic Stability* (Cambridge University Press, Cambridge, 2004)
3. P. Falsaperla, A. Giacobbe, G. Mulone, Nonlinear stability results for plane Couette and Poiseuille flows. *Phys. Rev. E* **100**(1), 013113 (2019)
4. P. Falsaperla, G. Mulone, C. Perrone, Energy stability of plane Couette and Poiseuille flows: a conjecture. *Eur. J. Mech.-B/Fluids* **93**, 93–100 (2022)
5. G. Mulone, Nonlinear monotone energy stability of plane shear flows: Joseph or Orr critical thresholds? *SIAM J. Appl. Math.* **84**(1), 60–74 (2024)
6. S.A. Orszag, Accurate solution of the Orr–Sommerfeld stability equation. *J. Fluid Mech.* **50**(4), 689–703 (1971)
7. J.J. Dongarra, B. Straughan, D.W. Walker, Chebyshev tau-QZ algorithm methods for calculating spectra of hydrodynamic stability problems. *Appl. Numer. Math.* **22**(4), 399–434 (1996)
8. B. Straughan, *The Energy Method, Stability, and Nonlinear Convection* (Springer, New York, 2013)
9. G. Arnone, J.A. Gianfrani, G. Massa, Chebyshev- $\tau$  method for certain generalized eigenvalue problems occurring in hydrodynamics: a concise survey. *Eur. Phys. J. Plus* **138**(3), 281 (2023)
10. K. Fujimura, R. Kelly, Interaction between longitudinal convection rolls and transverse waves in unstably stratified plane Poiseuille flow. *Phys. Fluids* **7**(1), 68–79 (1995)
11. Wolfram Research, Inc.: *Mathematica*, Version 13.3. Champaign, IL, 2023. <https://www.wolfram.com/mathematica>
12. A. Zebib, A Chebyshev method for the solution of boundary value problems. *J. Comput. Phys.* **53**(3), 443–455 (1984)